Spatial and temporal patterns of human avoidance by brown bears in a reintroduced population

Hystrix

Authors: Valentina Oberosler, Simone Tenan, Francesco Rovero

Corresponding author: Valentina Oberosler

MUSE - Museo delle Scienze, Corso del Lavoro e della Scienza, 3

38122 Trento, Italy

valentina.oberosler@muse.it

0039 0461 270374

Supplementary Data S1. R and JAGS code for the occupancy model, including the Bayesian variable selection approach, fitted to the detection/non-detection data of the brown bear in the central Italian Alps.

```
########### model

modelFilename = "Bear_GVS.txt"
cat("
    model {

    # priors for psi
    alpha0<- log(mean.psi/(1-mean.psi))
    mean.psi ~ dunif(0,1)

    # year effects (corner constraint)
    year.eff.psi[1] <- 0
    for(t in 2:nyear){
    year.eff.psi[t] ~ dunif(-5,5)
    }

    #year-specific intercepts
    intercept.psi_y1 <- mean.psi
    intercept.psi_y2 <- 1/(1+exp(-alpha0-year.eff.psi[2]))
    intercept.psi_y3 <- 1/(1+exp(-alpha0-year.eff.psi[3]))
    intercept.psi_y4 <- 1/(1+exp(-alpha0-year.eff.psi[4]))

    # priors for p
    beta0<- log(mean.p/(1-mean.p))
    mean.p ~ dunif(0,1)

    # year effects (corner constraint)
    year.eff.p[1] <- 0
    for(t in 2:nyear){
    year.eff.p[t] ~ dunif(-5,5)
    }
```

```
    intercept.p_y1 <- mean.p
    intercept.p_y2 <- 1/(1+exp(-beta0-year.eff.p[2]))
    intercept.p_y3 <- 1/(1+exp(-beta0-year.eff.p[3]))
    intercept.p_y4 <- 1/(1+exp(-beta0-year.eff.p[4]))

### effects (Link and Barker prior)
for(i in 1:6){
      predictorsIn[i] <- 1 + w[i]
      alpha[i] ~ dnorm(0,psuedoTau[i])T(-3,3)
      tauV[i] ~ dgamma(3.2890,7.8014)
      psuedoTau[i] <- ifelse(w[i]==1,
                             1/((1/tauV[i]) / predictorsIn[i]),
                             0.01*100)
      # Priors for variable indicators
      w[i] ~ dbern(0.5)
}

    for(i in 1:nsite){
    for(t in 1:nyear){
    tr_ped_st[i,t] ~ dunif(-10,10)
    tr_veh_sqrt_st[i,t] ~ dunif(-10,10)
    } }

    ## state process bear

    for (i in 1:nsite){
    for(t in 1:nyear){
    Z[i,t] ~ dbern(psi[i,t])
             logit(psi[i,t]) <- alpha0 + year.eff.psi[t]  + w[1] * alpha[1] *
             tr_ped_st[i,t] + w[2] * alpha[2] * tr_veh_sqrt_st[i,t]
             + w[3] * alpha[3] * Village_st[i] + w[4] * alpha[4] * Slope_st[i]
    } }

    ## observation process bear

    for(i in 1:nsite){
    for(t in 1:nyear){
    y_bear[i,t] ~ dbin(p[i,t] * Z[i,t],K_tot[i,t])

             logit(p[i,t]) <- beta0 + year.eff.p[t] + w[5] * alpha[5] * tr_ped_st[i,t]
             + w[6] * alpha[6] * Rotr[i]

    } }

    ## derived parameters

    for(t in 1:nyear){
    n_occ[t] <- sum(Z[,t])
    psi_mean[t] <- mean(psi[,t])
    p_mean[t] <- mean(p[,t])
    }
    }

    ## end model
    ", fill=TRUE, file=modelFilename)
```